
ПРИЛОЖЕНИЕ 3 

Технология создания списков 

 

При изучении списков MS Excel и баз данных MS Access возникает необходимость в создании 

разнообразных списков, причем в большинстве случаев в них используются не реальные, а некие ус-

ловные данные. Для эффективной работы со списками необходимо, чтобы они имели достаточное ко-

личество записей. Вводить же их вручную весьма трудоемко. Ниже описано, как с помощью не очень 

сложных формул в Microsoft Excel можно создать список практически любой конфигурации. Сформи-

рованный список затем можно экспортировать, например, в MS Access. 

Рассмотрим настройку рабочего листа MS Excel на примере создания списка «Отдел кадров» с 

полями: Фамилия, Имя, Отчество, Дата рождения, Код, Пол.  

Этапы работы:  

 

1. На рабочем листе ФИО (рис. 1) создайте базу фамилий (в мужском роде), имен (мужских и 

женских отдельно). 

 

 
Рис. 1. Лист ФИО 

 

2. С помощью формул (на рис. 1 они обозначены Ф1) сформируйте написание «женского» ва-

рианта фамилии.  Для этого в ячейке В3 запишите формулу:  

=ЕСЛИ(ЕТЕКСТ(A3);ЕСЛИ(ИЛИ(ПРАВСИМВ(A3;1)="в"; ПРАВСИМВ(A3;2)= "ин");A3&"а"; 

ЕСЛИ(ПРАВСИМВ(A3;2) ="ий"; ПОДСТАВИТЬ(A3; "ий";"ая"); ЕСЛИ(ПРАВСИМВ(A3;2) 

="ой";ПОДСТАВИТЬ(A3; "ой";"ая");A3)));"")  

Распространим (скопируем) ее на остальные ячейки столбца В. 

Примечание. Многообразие русских (и не только) фамилий не позволяет записать исчерпы-

вающую формулу. Учитывая условность записываемых в списки значений, данная и приведенные ни-

же формулы могут быть приняты. 

3. Сформируйте по списку мужских имен мужские и женские отчества.  Данный процесс 

представляет достаточные трудности, связанные с непростыми правилами русского языка и исключе-

ниями из правил. Запишем формулы для отчеств, образованных от наиболее распространенных рус-

ских имен для мужчин - формула Ф2 и для женщин - формула Ф3.  

Формула в ячейке Е3:  

=ЕСЛИ(ЕТЕКСТ(C3); 

ЕСЛИ(ПРАВСИМВ(C3;2)="ей";ПОДСТАВИТЬ(C3;"й";"евич"); 

ЕСЛИ(ПРАВСИМВ(C3;2)="ай";ПОДСТАВИТЬ(C3;"й";"евич"); 

ЕСЛИ(ПРАВСИМВ(C3;2)="ий";ПОДСТАВИТЬ(C3;"ий";"ьевич"); 

ЕСЛИ(ПРАВСИМВ(C3;1)="ь";ПОДСТАВИТЬ(C3;"ь";"евич"); 

ЕСЛИ(ПРАВСИМВ(C3;1)="а";ПОДСТАВИТЬ(C3;"а";"ич"); 



 

ЕСЛИ(ПРАВСИМВ(C2;1)="я";ПОДСТАВИТЬ(C2;"я";"ич"); 

СЦЕПИТЬ(C3;"ович")))))));"") 

Формула в ячейке F3:  

=ЕСЛИ(ЕТЕКСТ(E3); 

ЕСЛИ(ПРАВСИМВ(E3;1)="а";ПОДСТАВИТЬ(E3;"а";"ична"); 

ЕСЛИ(ПРАВСИМВ(E3;1)="я";ПОДСТАВИТЬ(E3;"я";"нична"); 

ПОДСТАВИТЬ(E3;"ич";"на")));"") 

Данные формулы обладают рядом существенных недостатков, главным из которых является то, 

что в ней не учитываются имена с бегающими гласными (Лев – Львович, Павел – Павлович), имена, 

заканчивающие на -ов, -аил (Яков – Яковлевич, Михаил – Михайлович), имена, заканчивающие на -ий, 

перед которым две согласные буквы (Дмитрий – Дмитриевич, Георгий – Георгиевич). 

Полученные на листе ФИО данные являются исходными для формирования списков. 

4. На листе Список получите случайным образом выбранные данные с полями Фамилия, Имя, 

Отчество, Дата рождения, Код, Пол (рис. 2). Обратите внимание на то, что для получения мужских и 

женских вариантов имен, отчеств, фамилий поле Пол является обязательным. 

 

 
Рис. 2. Лист Список 

 

Данные получим с помощью значений и формул:  

 диапазон В2:Н3 - начальные и конечные значения для формирования случайного числа из 

заданного диапазона; 

 ячейка В3- количество человек в списке (не более 500); 

 диапазон С2:Е2 - номер строки первого значения на рабочем листе ФИО; 

 Ф4 в ячейке С3 - номер строки после последнего непустого значения в столбце Фамилия на 

листе ФИО (не более 500):  

=C2+СЧЁТЕСЛИ(ФИО!A3:A503;">""") 

 Ф5 в ячейке D3 - номер строки после последнего непустого значения в столбце Имя / Муж-

чины на листе ФИО:  

=D2+СЧЁТЕСЛИ(ФИО!C3:C503;">""") 

 Ф6 в ячейке Е3 - номер строки после последнего непустого значения в столбце Имя / Жен-

щины на листе ФИО:  

=E2+СЧЁТЕСЛИ(ФИО!D3:D503;">""") 

 Ф7 в ячейке В6 - номер первого человека в списке: =B2 

Все нижеследующие формулы заполняют ячейку непустым значением только в том случае, ес-

ли ячейка с номером содержит число. После этого формула копируется до 503 строки 

 Ф8 в ячейке В7 - следующий номер, заполняется в том случае, если значение не превосходит 

верхнюю границу:  

=ЕСЛИ(B6<B$3;B6+1;"") 



 

 Ф9 в ячейке F6 - пол формируется случайным образом (при значении 1 в ячейку заносится 

значение «м», при 2 - «ж»)  

=ЕСЛИ(ЕЧИСЛО($B6);ВЫБОР(1+ЦЕЛОЕ(СЛЧИС()*2);"м";"ж");"") 

 Ф10 в ячейке С6 - в зависимости от выбранного пола выбирается фамилия случайным обра-

зом из одного из двух списков:  

=ЕСЛИ(ЕЧИСЛО($B6); ЕСЛИ($F6="м";  

ДВССЫЛ("ФИО!A"&ЦЕЛОЕ(C$2+СЛЧИС()*(C$3-C$2))); 

ДВССЫЛ("ФИО!B"&ЦЕЛОЕ(C$2+СЛЧИС()*(C$3-C$2))));"") 

 Ф11 в ячейке D6 - в зависимости от выбранного пола выбирается имя случайным образом из 

одного из двух списков:  

=ЕСЛИ(ЕЧИСЛО(B6);ЕСЛИ(F6="м"; 

ДВССЫЛ("ФИО!C"&ЦЕЛОЕ(D$2+СЛЧИС()*(D$3-D$2))); 

ДВССЫЛ("ФИО!D"&ЦЕЛОЕ(E$2+СЛЧИС()*(E$3-E$2))));"") 

 Ф12 в ячейке E6 - в зависимости от выбранного пола выбирается отчество случайным обра-

зом из одного из двух списков:  

=ЕСЛИ(ЕЧИСЛО(B6);ЕСЛИ(F6="м"; 

ДВССЫЛ("ФИО!E"&ЦЕЛОЕ(D$2+СЛЧИС()*(D$3-D$2))); 

ДВССЫЛ("ФИО!F"&ЦЕЛОЕ(D$2+СЛЧИС()*(D$3-D$2))));"") 

 Ф13 в ячейке G6 - случайное число из заданного диапазона, формат ячейки – Дата/время:

  

=ЕСЛИ(ЕЧИСЛО($B6);ЦЕЛОЕ(G$2+СЛЧИС()*(G$3-G$2+1));"") 

 Ф14 в ячейке H6 - случайное целое число из заданного диапазона:  

=ЕСЛИ(ЕЧИСЛО($B6); ЦЕЛОЕ(I$2+СЛЧИС()*(I$3-I$2+1));"") 

5. Распространите (скопируйте) формулы в каждом столбце – список готов (рис. 3).  

 

 
Рис. 3. Готовый список из 5 фамилий 

 

6. Выделите нужные поля (несмежные диапазоны выделяются с помощью клавиши CTRL) 

полученного списка и скопируйте на новый лист с помощью команды Специальная вставка, устано-

вив переключатель в положение Вставить значения. 

Аналогичным образом могут быть получены и другие данные (номера телефонов, даты приема 

на работу, оценки и т.п.). 

Полученный список можно использовать как в Excel, так и экспортировать в Access. 

